// File: contracts/core/VoxelTowers.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.4;

import "@openzeppelin/contracts/token/ERC721/ERC721.sol";
import "@openzeppelin/contracts/token/common/ERC2981.sol";
import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/utils/Base64.sol";
import "./IPropertyStore.sol";


/**
 * @title  Official VoxelTowers contract
 * @author The VoxelTowers crew
 * @notice The VoxelTowers contract implements updatable NFTs where each NFT represents ownership of an Apartment
 *         within the Voxel Towers (see https://www.voxel-towers.com)
 */
contract VoxelTowers is ERC721, ERC2981, Ownable, IPropertyStore {
    struct BasePropsParams {
        uint256 id;
        string properties;
        string imageLink;
    }

    struct BaseProps {
        uint256 tower;
        uint256 floor;
        string type_;
        string size;
        string view_;
        string direction;
        string balcony;
        string area;
        string ownership;
        string maisonette;
    }


    uint96 private royaltyPercentage = 1000; // 10.00%
    mapping(uint256 => IPropertyStore.Property[]) private tokensProps;
    mapping(uint256 => string) private tokensImage;
    mapping(address => bool) private propertySetters;


    constructor() ERC721("VoxelTowers", "VTO") {
        _setDefaultRoyalty(owner(), royaltyPercentage);
        mintTokens();
    }

    /**
     * @notice Mints all VTO tokens for all towers
     */
    function mintTokens() private {
        uint16[46] memory t1Apartments = getTower1TokenIds();
        for (uint256 i = 0; i < t1Apartments.length; i++) {
            _mint(owner(), t1Apartments[i]);
        }

        uint16[34] memory t2Apartments = getTower2TokenIds();
        for (uint256 i = 0; i < t2Apartments.length; i++) {
            _mint(owner(), t2Apartments[i]);
        }

        uint16[64] memory t3Apartments = getTower3TokenIds();
        for (uint256 i = 0; i < t3Apartments.length; i++) {
            _mint(owner(), t3Apartments[i]);
        }
    }

    /**
     * @notice Returns the token Ids of the tower 1 apartments
     * @return the tower 1 token Ids
     */
    function getTower1TokenIds() private pure returns (uint16[46] memory) {
        uint16[46] memory apartmentIds = [
            /* ids -> d[4]: tower, d[2-3]: floor, d[0-1]: apartment */
            // Floor 1
            10101, 10102, 10103, 10104,
            // Floor 2
            10201, 10202, 10203, 10204,
            // Floor 3
            10301, 10302, 10303, 10304,
            // Floor 4 (Lobby)
            10401, 10402,
            // Floor 5
            10501, 10502, 10503,
            // Floor 6
            10601, 10602, 10603,
            // Floor 7
            10701, 10702, 10703,
            // Floor 8
            10801, 10802, 10803,
            // Floor 9
            10901, 10902, 10903,
            // Floor 10
            11001, 11002, 11003,
            // Floor 11
            11101, 11102, 11103,
            // Floor 12
            11201, 11202, 11203,
            // Floor 13
            11301, 11302, 11303,
            // Floor 14
            11401, 11402, 11403,
            // Floor 15
            11501,
            // Floor 16
            11601
        ];
        return apartmentIds;
    }

    /**
     * @notice Returns the token Ids of the tower 2 apartments
     * @return the tower 2 token Ids
     */
    function getTower2TokenIds() private pure returns (uint16[34] memory) {
        uint16[34] memory apartmentIds = [
            /* ids -> d[4]: tower, d[2-3]: floor, d[0-1]: apartment */
            // Floor 1
            20101, 20102, 20103, 20104,
            // Floor 2
            20201, 20202, 20203, 20204,
            // Floor 3
            20301, 20302, 20303, 20304,
            // Floor 4 (Lobby)
            20401, 20402,
            // Floor 5
            20501, 20502, 20503,
            // Floor 6
            20601, 20602, 20603,
            // Floor 7
            20701, 20702, 20703,
            // Floor 8
            20801, 20802, 20803,
            // Floor 9
            20901, 20902, 20903,
            // Floor 10
            21001, 21002, 21003,
            // Floor 11
            21101,
            // Floor 12
            21201
        ];
        return apartmentIds;
    }

    /**
     * @notice Returns the token Ids of the tower 3 apartments
     * @return the tower 3 token Ids
     */
    function getTower3TokenIds() private pure returns (uint16[64] memory) {
        uint16[64] memory apartmentIds = [
            /* ids -> d[4]: tower, d[2-3]: floor, d[0-1]: apartment */
            // Floor 0 (Groundfloor)
            30001,
            // Floor 1
            30101, 30102, 30103, 30104, 30105, 30106, 30107, 30108, 30109, 30110, 30111, 30112,
            // Floor 2
            30201, 30202, 30203, 30204, 30205, 30206, 30207, 30208, 30209, 30210, 30211, 30212,
            // Floor 3
            30301, 30302, 30303, 30304, 30305, 30306, 30307, 30308, 30309, 30310, 30311, 30312,
            // Floor 4 (Lobby)
            30401, 30402, 30403, 30404,
            // Floor 5
            30501, 30502, 30503, 30504, 30505, 30506, 30507, 30508,
            // Floor 6
            30601, 30602, 30603, 30604, 30605, 30606, 30607, 30608,
            // Floor 7
            30701, 30702, 30703, 30704,
            // Floor 8
            30801, 30802,
            // Floor 9
            30901
        ];
        return apartmentIds;
    }


    /**
     * @notice Updates the image link of multiple tokens
     * @dev    Only the contract owner is allowed to update the image links
     * @param  tokenIds - the token Ids of the tokens
     * @param  imageLink - the new image link
     */
    function updateImageLink(uint256[] memory tokenIds, string memory imageLink) public onlyOwner {
        for (uint256 i = 0 ; i < tokenIds.length ; i++) {
            require(_exists(tokenIds[i]), "VoxelTowers: operator query for nonexistent token");
            tokensImage[tokenIds[i]] = imageLink;
        }
    }


    /**
     * @notice Sets base properties for all tokens in a tower
     * @dev    Only the contract owner is allowed to set the base properties. 
     *         This is a one time call for each tower
     * @param  tower - the tower Id
     * @param  basePropsParams - the base property parameters
     */
    function setBaseProperties(uint256 tower, BasePropsParams[] calldata basePropsParams) public onlyOwner {
        require(
            tower == 1 ? basePropsParams.length == 46 :
            tower == 2 ? basePropsParams.length == 34 :
            tower == 3 ? basePropsParams.length == 64 : false,
            "VoxelTowers: invalid parameter length"
        );
        
        for (uint256 i = 0; i < basePropsParams.length; i++) {
            require(_exists(basePropsParams[i].id), "VoxelTowers: operator query for nonexistent token");
            require(tokensProps[basePropsParams[i].id].length == 0, "VoxelTowers: initial properties already set");

            tokensProps[basePropsParams[i].id].push(IPropertyStore.Property("baseProps", basePropsParams[i].properties));
            tokensImage[basePropsParams[i].id] = basePropsParams[i].imageLink;
        }
    }


    /// @inheritdoc IERC165
    function supportsInterface(bytes4 interfaceId) public view virtual override(ERC721, ERC2981) returns (bool) {
        return super.supportsInterface(interfaceId);
    }


    /**
     * @notice Returns the token URI of a token
     * @param  tokenId - the token Id of the token
     * @return the token URI of the given token Id
     */
    function tokenURI(uint256 tokenId) public view override returns (string memory) {
        require(_exists(tokenId), "VoxelTowers: operator query for nonexistent token");
        BaseProps memory baseProps = getBaseProps(tokenId);

        bytes memory attributes = abi.encodePacked(
            '[',
                byteifyBaseProps(baseProps),
                byteifyAdditionalProps(tokenId),
            ']'
        );

        string memory json = Base64.encode(abi.encodePacked(
            '{'
                '"name":' '"', createTitle(tokenId, baseProps), '",'
                '"description":' '"', createDescription(tokenId, baseProps), '",'
                '"image":' '"', tokensImage[tokenId], '",'
                '"external_url":' '"', createLink(tokenId), '",'
                '"attributes":', attributes,
            '}'
        ));

        return string(abi.encodePacked('data:application/json;base64,', json));
    }

    /**
     * @notice Returns the base properties for a token
     * @param  tokenId - the token Id of the token
     * @return the base properties
     */
    function getBaseProps(uint256 tokenId) private view returns (BaseProps memory) {
        BaseProps memory props;
        props.tower = tokenId / 10000;
        props.floor = (tokenId / 100) % 100;
        
        string memory baseProps = tokensProps[tokenId][0].value;
        string[7] memory sizeProps = getSizeDerivedProps(tokenId, props.tower, props.floor, substring(baseProps, 0, 1));

        props.type_ = sizeProps[0];
        props.size = sizeProps[1];
        props.area = sizeProps[2];
        props.balcony = sizeProps[3];
        props.ownership = sizeProps[4];
        props.maisonette = sizeProps[5];
        props.view_ = sizeProps[6];
        props.direction = getDirection(substring(baseProps, 1, 2));

        return props;
    }

    /**
     * @notice Converts a string into a substring
     * @dev    See https://ethereum.stackexchange.com/questions/31457/substring-in-solidity
     * @param  str - the string
     * @param  startIndex - start index
     * @param  endIndex - end index
     * @return the substring
     */
    function substring(string memory str, uint256 startIndex, uint256 endIndex) private pure returns (string memory ) {
        bytes memory strBytes = bytes(str);
        bytes memory result = new bytes(endIndex - startIndex);
        for (uint256 i = startIndex; i < endIndex; i++) {
            result[i - startIndex] = strBytes[i];
        }
        return string(result);
    }

    /**
     * @notice Returns room size derived properties
     * @param  tokenId - the token Id
     * @param  tower - the tower Id
     * @param  floor - the floor number
     * @param  sizeStr - the size string
     * @return the size derived properties
     */
    function getSizeDerivedProps(uint256 tokenId, uint256 tower, uint256 floor, string memory sizeStr) private pure returns (string[7] memory) {
        // return [ <type>, <size>, <area>, <balcony>, <ownership>, <maisonette>, <view> ]
        string memory notAvailableString = "N/A";
        if (isSameString(sizeStr, "S")) {
            return [
                "Studio Apartment",
                "XS",
                "125",
                "No",
                "0.3",
                "No",
                "Standard"
            ];
        }
        if (isSameString(sizeStr, "s")) {
            if (tokenId == 10401 || tokenId == 20401) {
                return [
                    "Commercial",
                    "S",
                    "175",
                    "No",
                    "0",
                    "No",
                    notAvailableString
                ];
            } else if (tokenId == 10402 || tokenId == 20402) {
                return [
                    "Community",
                    "S",
                    "175",
                    "No",
                    "0",
                    "No",
                    notAvailableString
                ];
            } else {
                return [
                    "Corner Apartment",
                    "S",
                    "175",
                    "No",
                    "0.5",
                    "No",
                    "90"
                ];
            }
        }
        if (isSameString(sizeStr, "M")) {
            if (tower == 3 && floor == 4) {
                return [
                    "Community",
                    "XM",
                    "200",
                    "No",
                    "0",
                    "No",
                    notAvailableString
                ];
            } else {
                return [
                    "Superior Apartment",
                    "XM",
                    "200",
                    "1 Small",
                    "0.5",
                    "Yes",
                    "Standard"
                ];
            }
        }
        if (isSameString(sizeStr, "m")) {
            return [
                "Junior Suite",
                "M",
                "250",
                "1 Medium",
                "0.7",
                "Yes",
                "90"
            ];
        }
        if (isSameString(sizeStr, "l")) {
            if (tower == 3 && floor == 0) {
                return [
                    "Commercial",
                    "L",
                    "1000",
                    "No",
                    "0",
                    "No",
                    notAvailableString
                ];
            } else {
                return [
                    "Executive Suite",
                    "L",
                    "500",
                    "1 Large",
                    "1.3",
                    "Yes",
                    "180"
                ];
            }
        }
        if (isSameString(sizeStr, "o")) {
            return [
                tower == 1 ? "The Voxel Loft" :
                tower == 2 ? "The Parcel Loft" :
                tower == 3 ? "The $SAND Loft" : "",
                "Loft",
                "1000",
                "2 Large",
                "2.6",
                "Yes",
                "360"
            ];
        }
        if (isSameString(sizeStr, "p")) {
            return [
                tower == 1 ? "The Arthur Penthouse" :
                tower == 2 ? "The Borget Penthouse" : "",
                "Penthouse",
                "1250",
                "2 Large",
                "3.3",
                "Yes",
                "360"
            ];
        }
        return ["", "", "", "", "", "", ""];
    }

    /**
     * @notice Checks if two strings are the same
     * @dev See https://ethereum.stackexchange.com/questions/30912/how-to-compare-strings-in-solidity
     * @param  str1 - string 1
     * @param  str2 - string 2
     * @return true if strings are the same, false if not
     */
    function isSameString(string memory str1, string memory str2) private pure returns (bool) {
        return keccak256(bytes(str1)) == keccak256(bytes(str2));
    }

    /**
     * @notice Concats two strings
     * @param  baseStr - base string
     * @param  appendStr - append string
     * @return the concatinated string
     */
    function concatString(string memory baseStr, string memory appendStr) private pure returns (string memory) {
        return string(abi.encodePacked(baseStr, appendStr));
    }

    /**
     * @notice Returns the room direction
     * @param  directionStr - the direction string
     * @return the room direction
     */
    function getDirection(string memory directionStr) private pure returns (string memory) {
        if (isSameString(directionStr, "c")) {
            return "Center";
        }
        if (isSameString(directionStr, "o")) {
            return "Outside";
        }
        if (isSameString(directionStr, "p")) {
            return "Panorama";
        }
        return "N/A";
    }

     /**
     * @notice Formats and byteifies the provided base properties
     * @param  baseProps - the base properties
     * @return the byteified base properties
     */
    function byteifyBaseProps(BaseProps memory baseProps) private pure returns (bytes memory) {
        bytes memory propsBytes = abi.encodePacked(
            '{'
                '"trait_type":' '"Tower",'
                '"value":' '"', Strings.toString(baseProps.tower), '"'
            '},'
        );
        propsBytes = abi.encodePacked(propsBytes, abi.encodePacked(
            '{'
                '"trait_type":' '"Floor",'
                '"value":' '"', Strings.toString(baseProps.floor), '"'
            '},'
        ));
        propsBytes = abi.encodePacked(propsBytes, abi.encodePacked(
            '{'
                '"trait_type":' '"Type",'
                '"value":' '"', baseProps.type_, '"'
            '},'
        ));
        propsBytes = abi.encodePacked(propsBytes, abi.encodePacked(
            '{'
                '"trait_type":' '"Size",'
                '"value":' '"', baseProps.size, '"'
            '},'
        ));
        propsBytes = abi.encodePacked(propsBytes, abi.encodePacked(
            '{'
                '"trait_type":' '"Area [m2]",'
                '"value":' '"', baseProps.area, '"'
            '},'
        ));
        propsBytes = abi.encodePacked(propsBytes, abi.encodePacked(
            '{'
                '"trait_type":' '"Maisonette",'
                '"value":' '"', baseProps.maisonette, '"'
            '},'
        ));
        propsBytes = abi.encodePacked(propsBytes, abi.encodePacked(
            '{'
                '"trait_type":' '"Balcony",'
                '"value":' '"', baseProps.balcony, '"'
            '},'
        ));
        propsBytes = abi.encodePacked(propsBytes, abi.encodePacked(
            '{'
                '"trait_type":' '"View [deg]",'
                '"value":' '"', baseProps.view_, '"'
            '},'
        ));
        propsBytes = abi.encodePacked(propsBytes, abi.encodePacked(
            '{'
                '"trait_type":' '"Direction",'
                '"value":' '"', baseProps.direction, '"'
            '},'
        ));
        propsBytes = abi.encodePacked(propsBytes, abi.encodePacked(
            '{'
                '"trait_type":' '"Location",'
                '"value":' '"LAND (147, 96)"'
            '},'
        ));
        propsBytes = abi.encodePacked(propsBytes, abi.encodePacked(
            '{'
                '"trait_type":' '"Metaverse",'
                '"value":' '"The Sandbox"'
            '},'
        ));
        propsBytes = abi.encodePacked(propsBytes, abi.encodePacked(
            '{'
                '"trait_type":' '"Land Ownership [%]",'
                '"value":' '"', baseProps.ownership, '"'
            '}'
        ));

        return propsBytes;
    }

    /**
     * @notice Formats and byteifies the additional properties
     * @param  tokenId - the token Id
     * @return the byteified additional properties
     */
    function byteifyAdditionalProps(uint256 tokenId) private view returns (bytes memory) {
        bytes memory propsBytes = "";

        for (uint256 i = 1 ; i < tokensProps[tokenId].length ; i++) {
            propsBytes = abi.encodePacked(propsBytes, abi.encodePacked(
                ',{'
                    '"trait_type":' '"', tokensProps[tokenId][i].key, '",'
                    '"value":' '"', tokensProps[tokenId][i].value, '"'
                '}'
            ));
        }

        return propsBytes;
    }

    /**
     * @notice Creates the title of a token
     * @param  tokenId - the token Id
     * @param  baseProps - the base properties
     * @return the title of the token
     */
    function createTitle(uint256 tokenId, BaseProps memory baseProps) private pure returns (string memory) {
        // Corner Apartment #10101 (Tower 1, Floor 1)
        return string(abi.encodePacked(
            baseProps.type_,
            ' #',
            Strings.toString(tokenId),
            ' (Tower ',
            Strings.toString(baseProps.tower),
            ', Floor ',
            Strings.toString(baseProps.floor),
            ')'
        ));
    }

    /**
     * @notice Creates the description of a token
     * @param  tokenId - the token Id
     * @param  baseProps - the base properties
     * @return the description of the token
     */
    function createDescription(uint256 tokenId, BaseProps memory baseProps) private pure returns (string memory) {
        // This NFT represents ownership of Corner Apartment #10101 in the Voxel Towers.
        return string (abi.encodePacked(
            'This NFT represents ownership of ',
            baseProps.type_,
            ' #',
            Strings.toString(tokenId),
            ' in the Voxel Towers.'
        ));
    }

    /**
     * @notice Returns the link of a token
     * @param  tokenId - the token Id
     * @return the link of the token
     */
    function createLink(uint256 tokenId) private pure returns (string memory) {
        // https://www.voxel-towers.com/apartment/?tokenId=10101
        return string (abi.encodePacked(
            'https://www.voxel-towers.com/apartment/?tokenId=',
            Strings.toString(tokenId)
        ));
    }


    /**
     * @notice Updates an address to permit/deny to set additional properties
     * @dev    This function is only allowed to be called by the contract owner.
     *         Per default every address is denied to add properties.
     *         To permit an address, call this function to toggle between permit/deny states.
     * @param  address_ - the address
     */
    function updatePropertySetters(address address_) public onlyOwner {
        propertySetters[address_] = !propertySetters[address_];
    }


    /**
     * @notice Checks if an address is permitted to set additional properties
     * @param  address_ - the address
     * @return true if address is permitted, false if not
     */
    function isPropertySetter(address address_) public view returns (bool) {
        return propertySetters[address_];
    }


    /// @inheritdoc IPropertyStore
    function addProperty(uint256 tokenId, IPropertyStore.Property calldata property) public virtual override {
        require(_exists(tokenId), "VoxelTowers: operator query for nonexistent token");
        require(propertySetters[msg.sender], "VoxelTowers: sender not whitelisted");

        for (uint256 i = 0; i < tokensProps[tokenId].length; i++) {
            require(!isSameString(tokensProps[tokenId][i].key, property.key), "VoxelTowers: property already set");
        }
        tokensProps[tokenId].push(property);
    }


    /// @inheritdoc IPropertyStore
    function getProperties(uint256 tokenId) public view virtual override returns (IPropertyStore.Property[] memory) {
        require(_exists(tokenId), "VoxelTowers: operator query for nonexistent token");
        return tokensProps[tokenId];
    }
}
// File: @openzeppelin/contracts/token/ERC721/ERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (token/ERC721/ERC721.sol)

pragma solidity ^0.8.0;

import "./IERC721.sol";
import "./IERC721Receiver.sol";
import "./extensions/IERC721Metadata.sol";
import "../../utils/Address.sol";
import "../../utils/Context.sol";
import "../../utils/Strings.sol";
import "../../utils/introspection/ERC165.sol";

/**
 * @dev Implementation of https://eips.ethereum.org/EIPS/eip-721[ERC721] Non-Fungible Token Standard, including
 * the Metadata extension, but not including the Enumerable extension, which is available separately as
 * {ERC721Enumerable}.
 */
contract ERC721 is Context, ERC165, IERC721, IERC721Metadata {
    using Address for address;
    using Strings for uint256;

    // Token name
    string private _name;

    // Token symbol
    string private _symbol;

    // Mapping from token ID to owner address
    mapping(uint256 => address) private _owners;

    // Mapping owner address to token count
    mapping(address => uint256) private _balances;

    // Mapping from token ID to approved address
    mapping(uint256 => address) private _tokenApprovals;

    // Mapping from owner to operator approvals
    mapping(address => mapping(address => bool)) private _operatorApprovals;

    /**
     * @dev Initializes the contract by setting a `name` and a `symbol` to the token collection.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(ERC165, IERC165) returns (bool) {
        return
            interfaceId == type(IERC721).interfaceId ||
            interfaceId == type(IERC721Metadata).interfaceId ||
            super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {IERC721-balanceOf}.
     */
    function balanceOf(address owner) public view virtual override returns (uint256) {
        require(owner != address(0), "ERC721: balance query for the zero address");
        return _balances[owner];
    }

    /**
     * @dev See {IERC721-ownerOf}.
     */
    function ownerOf(uint256 tokenId) public view virtual override returns (address) {
        address owner = _owners[tokenId];
        require(owner != address(0), "ERC721: owner query for nonexistent token");
        return owner;
    }

    /**
     * @dev See {IERC721Metadata-name}.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev See {IERC721Metadata-symbol}.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev See {IERC721Metadata-tokenURI}.
     */
    function tokenURI(uint256 tokenId) public view virtual override returns (string memory) {
        require(_exists(tokenId), "ERC721Metadata: URI query for nonexistent token");

        string memory baseURI = _baseURI();
        return bytes(baseURI).length > 0 ? string(abi.encodePacked(baseURI, tokenId.toString())) : "";
    }

    /**
     * @dev Base URI for computing {tokenURI}. If set, the resulting URI for each
     * token will be the concatenation of the `baseURI` and the `tokenId`. Empty
     * by default, can be overriden in child contracts.
     */
    function _baseURI() internal view virtual returns (string memory) {
        return "";
    }

    /**
     * @dev See {IERC721-approve}.
     */
    function approve(address to, uint256 tokenId) public virtual override {
        address owner = ERC721.ownerOf(tokenId);
        require(to != owner, "ERC721: approval to current owner");

        require(
            _msgSender() == owner || isApprovedForAll(owner, _msgSender()),
            "ERC721: approve caller is not owner nor approved for all"
        );

        _approve(to, tokenId);
    }

    /**
     * @dev See {IERC721-getApproved}.
     */
    function getApproved(uint256 tokenId) public view virtual override returns (address) {
        require(_exists(tokenId), "ERC721: approved query for nonexistent token");

        return _tokenApprovals[tokenId];
    }

    /**
     * @dev See {IERC721-setApprovalForAll}.
     */
    function setApprovalForAll(address operator, bool approved) public virtual override {
        _setApprovalForAll(_msgSender(), operator, approved);
    }

    /**
     * @dev See {IERC721-isApprovedForAll}.
     */
    function isApprovedForAll(address owner, address operator) public view virtual override returns (bool) {
        return _operatorApprovals[owner][operator];
    }

    /**
     * @dev See {IERC721-transferFrom}.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        //solhint-disable-next-line max-line-length
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: transfer caller is not owner nor approved");

        _transfer(from, to, tokenId);
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        safeTransferFrom(from, to, tokenId, "");
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) public virtual override {
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: transfer caller is not owner nor approved");
        _safeTransfer(from, to, tokenId, _data);
    }

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * `_data` is additional data, it has no specified format and it is sent in call to `to`.
     *
     * This internal function is equivalent to {safeTransferFrom}, and can be used to e.g.
     * implement alternative mechanisms to perform token transfer, such as signature-based.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeTransfer(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) internal virtual {
        _transfer(from, to, tokenId);
        require(_checkOnERC721Received(from, to, tokenId, _data), "ERC721: transfer to non ERC721Receiver implementer");
    }

    /**
     * @dev Returns whether `tokenId` exists.
     *
     * Tokens can be managed by their owner or approved accounts via {approve} or {setApprovalForAll}.
     *
     * Tokens start existing when they are minted (`_mint`),
     * and stop existing when they are burned (`_burn`).
     */
    function _exists(uint256 tokenId) internal view virtual returns (bool) {
        return _owners[tokenId] != address(0);
    }

    /**
     * @dev Returns whether `spender` is allowed to manage `tokenId`.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function _isApprovedOrOwner(address spender, uint256 tokenId) internal view virtual returns (bool) {
        require(_exists(tokenId), "ERC721: operator query for nonexistent token");
        address owner = ERC721.ownerOf(tokenId);
        return (spender == owner || getApproved(tokenId) == spender || isApprovedForAll(owner, spender));
    }

    /**
     * @dev Safely mints `tokenId` and transfers it to `to`.
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeMint(address to, uint256 tokenId) internal virtual {
        _safeMint(to, tokenId, "");
    }

    /**
     * @dev Same as {xref-ERC721-_safeMint-address-uint256-}[`_safeMint`], with an additional `data` parameter which is
     * forwarded in {IERC721Receiver-onERC721Received} to contract recipients.
     */
    function _safeMint(
        address to,
        uint256 tokenId,
        bytes memory _data
    ) internal virtual {
        _mint(to, tokenId);
        require(
            _checkOnERC721Received(address(0), to, tokenId, _data),
            "ERC721: transfer to non ERC721Receiver implementer"
        );
    }

    /**
     * @dev Mints `tokenId` and transfers it to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {_safeMint} whenever possible
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - `to` cannot be the zero address.
     *
     * Emits a {Transfer} event.
     */
    function _mint(address to, uint256 tokenId) internal virtual {
        require(to != address(0), "ERC721: mint to the zero address");
        require(!_exists(tokenId), "ERC721: token already minted");

        _beforeTokenTransfer(address(0), to, tokenId);

        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(address(0), to, tokenId);

        _afterTokenTransfer(address(0), to, tokenId);
    }

    /**
     * @dev Destroys `tokenId`.
     * The approval is cleared when the token is burned.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     *
     * Emits a {Transfer} event.
     */
    function _burn(uint256 tokenId) internal virtual {
        address owner = ERC721.ownerOf(tokenId);

        _beforeTokenTransfer(owner, address(0), tokenId);

        // Clear approvals
        _approve(address(0), tokenId);

        _balances[owner] -= 1;
        delete _owners[tokenId];

        emit Transfer(owner, address(0), tokenId);

        _afterTokenTransfer(owner, address(0), tokenId);
    }

    /**
     * @dev Transfers `tokenId` from `from` to `to`.
     *  As opposed to {transferFrom}, this imposes no restrictions on msg.sender.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     *
     * Emits a {Transfer} event.
     */
    function _transfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {
        require(ERC721.ownerOf(tokenId) == from, "ERC721: transfer from incorrect owner");
        require(to != address(0), "ERC721: transfer to the zero address");

        _beforeTokenTransfer(from, to, tokenId);

        // Clear approvals from the previous owner
        _approve(address(0), tokenId);

        _balances[from] -= 1;
        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(from, to, tokenId);

        _afterTokenTransfer(from, to, tokenId);
    }

    /**
     * @dev Approve `to` to operate on `tokenId`
     *
     * Emits a {Approval} event.
     */
    function _approve(address to, uint256 tokenId) internal virtual {
        _tokenApprovals[tokenId] = to;
        emit Approval(ERC721.ownerOf(tokenId), to, tokenId);
    }

    /**
     * @dev Approve `operator` to operate on all of `owner` tokens
     *
     * Emits a {ApprovalForAll} event.
     */
    function _setApprovalForAll(
        address owner,
        address operator,
        bool approved
    ) internal virtual {
        require(owner != operator, "ERC721: approve to caller");
        _operatorApprovals[owner][operator] = approved;
        emit ApprovalForAll(owner, operator, approved);
    }

    /**
     * @dev Internal function to invoke {IERC721Receiver-onERC721Received} on a target address.
     * The call is not executed if the target address is not a contract.
     *
     * @param from address representing the previous owner of the given token ID
     * @param to target address that will receive the tokens
     * @param tokenId uint256 ID of the token to be transferred
     * @param _data bytes optional data to send along with the call
     * @return bool whether the call correctly returned the expected magic value
     */
    function _checkOnERC721Received(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) private returns (bool) {
        if (to.isContract()) {
            try IERC721Receiver(to).onERC721Received(_msgSender(), from, tokenId, _data) returns (bytes4 retval) {
                return retval == IERC721Receiver.onERC721Received.selector;
            } catch (bytes memory reason) {
                if (reason.length == 0) {
                    revert("ERC721: transfer to non ERC721Receiver implementer");
                } else {
                    assembly {
                        revert(add(32, reason), mload(reason))
                    }
                }
            }
        } else {
            return true;
        }
    }

    /**
     * @dev Hook that is called before any token transfer. This includes minting
     * and burning.
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, ``from``'s `tokenId` will be
     * transferred to `to`.
     * - When `from` is zero, `tokenId` will be minted for `to`.
     * - When `to` is zero, ``from``'s `tokenId` will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {}

    /**
     * @dev Hook that is called after any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _afterTokenTransfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {}
}

// File: @openzeppelin/contracts/token/common/ERC2981.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (token/common/ERC2981.sol)

pragma solidity ^0.8.0;

import "../../interfaces/IERC2981.sol";
import "../../utils/introspection/ERC165.sol";

/**
 * @dev Implementation of the NFT Royalty Standard, a standardized way to retrieve royalty payment information.
 *
 * Royalty information can be specified globally for all token ids via {_setDefaultRoyalty}, and/or individually for
 * specific token ids via {_setTokenRoyalty}. The latter takes precedence over the first.
 *
 * Royalty is specified as a fraction of sale price. {_feeDenominator} is overridable but defaults to 10000, meaning the
 * fee is specified in basis points by default.
 *
 * IMPORTANT: ERC-2981 only specifies a way to signal royalty information and does not enforce its payment. See
 * https://eips.ethereum.org/EIPS/eip-2981#optional-royalty-payments[Rationale] in the EIP. Marketplaces are expected to
 * voluntarily pay royalties together with sales, but note that this standard is not yet widely supported.
 *
 * _Available since v4.5._
 */
abstract contract ERC2981 is IERC2981, ERC165 {
    struct RoyaltyInfo {
        address receiver;
        uint96 royaltyFraction;
    }

    RoyaltyInfo private _defaultRoyaltyInfo;
    mapping(uint256 => RoyaltyInfo) private _tokenRoyaltyInfo;

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(IERC165, ERC165) returns (bool) {
        return interfaceId == type(IERC2981).interfaceId || super.supportsInterface(interfaceId);
    }

    /**
     * @inheritdoc IERC2981
     */
    function royaltyInfo(uint256 _tokenId, uint256 _salePrice)
        external
        view
        virtual
        override
        returns (address, uint256)
    {
        RoyaltyInfo memory royalty = _tokenRoyaltyInfo[_tokenId];

        if (royalty.receiver == address(0)) {
            royalty = _defaultRoyaltyInfo;
        }

        uint256 royaltyAmount = (_salePrice * royalty.royaltyFraction) / _feeDenominator();

        return (royalty.receiver, royaltyAmount);
    }

    /**
     * @dev The denominator with which to interpret the fee set in {_setTokenRoyalty} and {_setDefaultRoyalty} as a
     * fraction of the sale price. Defaults to 10000 so fees are expressed in basis points, but may be customized by an
     * override.
     */
    function _feeDenominator() internal pure virtual returns (uint96) {
        return 10000;
    }

    /**
     * @dev Sets the royalty information that all ids in this contract will default to.
     *
     * Requirements:
     *
     * - `receiver` cannot be the zero address.
     * - `feeNumerator` cannot be greater than the fee denominator.
     */
    function _setDefaultRoyalty(address receiver, uint96 feeNumerator) internal virtual {
        require(feeNumerator <= _feeDenominator(), "ERC2981: royalty fee will exceed salePrice");
        require(receiver != address(0), "ERC2981: invalid receiver");

        _defaultRoyaltyInfo = RoyaltyInfo(receiver, feeNumerator);
    }

    /**
     * @dev Removes default royalty information.
     */
    function _deleteDefaultRoyalty() internal virtual {
        delete _defaultRoyaltyInfo;
    }

    /**
     * @dev Sets the royalty information for a specific token id, overriding the global default.
     *
     * Requirements:
     *
     * - `tokenId` must be already minted.
     * - `receiver` cannot be the zero address.
     * - `feeNumerator` cannot be greater than the fee denominator.
     */
    function _setTokenRoyalty(
        uint256 tokenId,
        address receiver,
        uint96 feeNumerator
    ) internal virtual {
        require(feeNumerator <= _feeDenominator(), "ERC2981: royalty fee will exceed salePrice");
        require(receiver != address(0), "ERC2981: Invalid parameters");

        _tokenRoyaltyInfo[tokenId] = RoyaltyInfo(receiver, feeNumerator);
    }

    /**
     * @dev Resets royalty information for the token id back to the global default.
     */
    function _resetTokenRoyalty(uint256 tokenId) internal virtual {
        delete _tokenRoyaltyInfo[tokenId];
    }
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/utils/Base64.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (utils/Base64.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides a set of functions to operate with Base64 strings.
 *
 * _Available since v4.5._
 */
library Base64 {
    /**
     * @dev Base64 Encoding/Decoding Table
     */
    string internal constant _TABLE = "ABCDEFGHIJKLMNOPQRSTUVWXYZabcdefghijklmnopqrstuvwxyz0123456789+/";

    /**
     * @dev Converts a `bytes` to its Bytes64 `string` representation.
     */
    function encode(bytes memory data) internal pure returns (string memory) {
        /**
         * Inspired by Brecht Devos (Brechtpd) implementation - MIT licence
         * https://github.com/Brechtpd/base64/blob/e78d9fd951e7b0977ddca77d92dc85183770daf4/base64.sol
         */
        if (data.length == 0) return "";

        // Loads the table into memory
        string memory table = _TABLE;

        // Encoding takes 3 bytes chunks of binary data from `bytes` data parameter
        // and split into 4 numbers of 6 bits.
        // The final Base64 length should be `bytes` data length multiplied by 4/3 rounded up
        // - `data.length + 2`  -> Round up
        // - `/ 3`              -> Number of 3-bytes chunks
        // - `4 *`              -> 4 characters for each chunk
        string memory result = new string(4 * ((data.length + 2) / 3));

        assembly {
            // Prepare the lookup table (skip the first "length" byte)
            let tablePtr := add(table, 1)

            // Prepare result pointer, jump over length
            let resultPtr := add(result, 32)

            // Run over the input, 3 bytes at a time
            for {
                let dataPtr := data
                let endPtr := add(data, mload(data))
            } lt(dataPtr, endPtr) {

            } {
                // Advance 3 bytes
                dataPtr := add(dataPtr, 3)
                let input := mload(dataPtr)

                // To write each character, shift the 3 bytes (18 bits) chunk
                // 4 times in blocks of 6 bits for each character (18, 12, 6, 0)
                // and apply logical AND with 0x3F which is the number of
                // the previous character in the ASCII table prior to the Base64 Table
                // The result is then added to the table to get the character to write,
                // and finally write it in the result pointer but with a left shift
                // of 256 (1 byte) - 8 (1 ASCII char) = 248 bits

                mstore8(resultPtr, mload(add(tablePtr, and(shr(18, input), 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance

                mstore8(resultPtr, mload(add(tablePtr, and(shr(12, input), 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance

                mstore8(resultPtr, mload(add(tablePtr, and(shr(6, input), 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance

                mstore8(resultPtr, mload(add(tablePtr, and(input, 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance
            }

            // When data `bytes` is not exactly 3 bytes long
            // it is padded with `=` characters at the end
            switch mod(mload(data), 3)
            case 1 {
                mstore8(sub(resultPtr, 1), 0x3d)
                mstore8(sub(resultPtr, 2), 0x3d)
            }
            case 2 {
                mstore8(sub(resultPtr, 1), 0x3d)
            }
        }

        return result;
    }
}

// File: contracts/core/IPropertyStore.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.4;

/**
 * @title  VoxelTowers Property interface
 * @author The VoxelTowers crew
 * @notice This interface defines the functions for external contracts to set additional properties to VTO tokens
 */
interface IPropertyStore {
    struct Property {
        string key;
        string value;
    }

    /**
     * @notice Adds an aditional property to a token
     * @dev    The caller address must be whitelisted first in order to add additional properties
     * @param  tokenId - the token Id
     * @param  property - the additional property
     */
    function addProperty(uint256 tokenId, Property calldata property) external;

    /**
     * @notice Gets the properties of a token
     * @param  tokenId - the token Id
     * @return the properties of token
     */
    function getProperties(uint256 tokenId) external view returns (Property[] memory);
}

// File: @openzeppelin/contracts/token/ERC721/IERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/IERC721.sol)

pragma solidity ^0.8.0;

import "../../utils/introspection/IERC165.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {safeTransferFrom} whenever possible.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool _approved) external;

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes calldata data
    ) external;
}

// File: @openzeppelin/contracts/token/ERC721/IERC721Receiver.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/IERC721Receiver.sol)

pragma solidity ^0.8.0;

/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to confirm the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721.onERC721Received.selector`.
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4);
}

// File: @openzeppelin/contracts/token/ERC721/extensions/IERC721Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/extensions/IERC721Metadata.sol)

pragma solidity ^0.8.0;

import "../IERC721.sol";

/**
 * @title ERC-721 Non-Fungible Token Standard, optional metadata extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Metadata is IERC721 {
    /**
     * @dev Returns the token collection name.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the token collection symbol.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the Uniform Resource Identifier (URI) for `tokenId` token.
     */
    function tokenURI(uint256 tokenId) external view returns (string memory);
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/utils/Strings.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Strings.sol)

pragma solidity ^0.8.0;

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _HEX_SYMBOLS = "0123456789abcdef";

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT licence
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        if (value == 0) {
            return "0x00";
        }
        uint256 temp = value;
        uint256 length = 0;
        while (temp != 0) {
            length++;
            temp >>= 8;
        }
        return toHexString(value, length);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _HEX_SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }
}

// File: @openzeppelin/contracts/utils/introspection/ERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/ERC165.sol)

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}

// File: @openzeppelin/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

// File: @openzeppelin/contracts/interfaces/IERC2981.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (interfaces/IERC2981.sol)

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Interface for the NFT Royalty Standard.
 *
 * A standardized way to retrieve royalty payment information for non-fungible tokens (NFTs) to enable universal
 * support for royalty payments across all NFT marketplaces and ecosystem participants.
 *
 * _Available since v4.5._
 */
interface IERC2981 is IERC165 {
    /**
     * @dev Returns how much royalty is owed and to whom, based on a sale price that may be denominated in any unit of
     * exchange. The royalty amount is denominated and should be payed in that same unit of exchange.
     */
    function royaltyInfo(uint256 tokenId, uint256 salePrice)
        external
        view
        returns (address receiver, uint256 royaltyAmount);
}

// File: @openzeppelin/contracts/interfaces/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (interfaces/IERC165.sol)

pragma solidity ^0.8.0;

import "../utils/introspection/IERC165.sol";